Canadian Mineralogist
Vol. 29, pp. 287-299 (1991)

A SIMPLE QUANTITATIVE CALCULATION OF MOL FRACTIONS OF AMPHIBOLE END-MEMBERS

KENNETH L. CURRIE
Geological Survey of Canada, 601 Booth Street, Ottawa, Ontario K1A 0E8

ABSTRACT

An amphibole composition space defined by *A*, *B* and *T* site occupancies subdivides into 13 tetrahedral domains, each with vertices defined by International Mineralogical Association end-members. With the assumption that a given composition comprises only end members of the tetrahedral domain in which it plots, mol fractions of such end-members can be assigned for any sample whose composition falls within the composition space. The most abundant end member calculated by this method commonly, but not invariably, agrees with the name assigned by IMA rules. The calculation is relatively insensitive to lack of water and ferric iron determinations. A computer program in BASIC is listed. The method can be applied to any complex solid-solution.

Keywords: amphibole, classification, end members, mol fractions, computer program.

SOMMAIRE

Une classification des amphiboles peut se faire dans l'espace compositionnel défini par l'occupation des sites *A*, *B* et *T*. Il en résulte treize domaines tétraédriques distincts, chacun ayant comme sommets des pôles approuvés par l'Association Internationale de Minéralogie. Si la supposition est faite qu'une composition se décompose uniquement en termes du domaine tétraédrique pertinent, il est possible de calculer les fractions molaires des pôles du domaine en question. Le pôle le plus important qui ressort du calcul correspond couramment, mais pas nécessairement, au nom qui résulte de l'application des règles de l'IMA. Le calcul est relativement insensible à l'absence de données sur la concentration de l'eau ou du fer trivalent. Le logiciel requis pour effectuer le calcul, écrit en BASIC, est inclus. La méthode peut s'appliquer à n'importe quelle solution solide complexe.

(Traduit par la Rédaction)

Mots-clés: amphibole, classification, pôles, fractions molaires, logiciel.

INTRODUCTION

Amphiboles form an enormously diverse mineral family, critical to the understanding of a great variety of igneous and metamorphic rocks. An amphibole can be named according to conventions established by the IMA (Leake 1978, Rock & Leake 1983). However, in a group of related amphibole compositions, or a series of methods of recalculation of one composition, formal names often vary widely, concealing the kinship between closely related compositions (*cf.* Richard & Clarke 1990). Analytical results may be directly and informatively compared *via* cation content of the formula unit, but quantitative relation of this unit to formal end-members requires some method of assignment to end members.

PRINCIPLES OF ASSIGNMENT TO END MEMBERS

Amphiboles have the ideal formula $A_{0\text{-}1}B_2C_5T_8O_{22}(OH,F,Cl)_2$, where commonly Na and K occupy the *A* site, Na, Li, Ca, Mn, Mg and Fe^{2+}, the *B* position, Mg, Fe^{2+}, Mn, Al, Fe^{3+} and Ti, the *C* position, and Si and Al, the *T* position, although many other less common substitutions occur. Assume that an amphibole composition has been expressed in this form, and a set of end members selected. (Neither of these problems is trivial, but for the present purpose they are assumed to be solved.) Can a meaningful mol fraction of a given end-member be calculated? Consider occupancy of the *A*, *B* and *T* sites. Independent variation at the *A*, *B* and *T* sites can easily be demonstrated by considering possible substitutions involving the *C* site, for example

$$v_A + M^{3+}_C = M^{1+}_A + M^{2+}_C$$
(riebeckite – arfvedsonite)

$$M^{2+}_B + M^{2+}_C = M^{1+}_B + M^{3+}_C$$
(tremolite – winchite)

$$M^{4+}_T + M^{2+}_C = M^{3+}_T + M^{3+}_C$$
(tremolite – hornblende)

where subscripts refer to the site, superscripts to the valence of the atom, *M* to any atom, and *v* to a vacancy. In parentheses is listed an example of the effect of such a substitution. Thus in the first case, riebeckite has the *A* site vacant, and a trivalent ion (ferric iron) in *C*. Substitution of a monovalent sodium in *A* and divalent iron for the trivalent iron in *C* gives arfvedsonite. Given the independence of *A*, *B* and *T*, an amphibole composition space based upon them can be constructed as shown in Figure 1, in which the dimensions have been chosen as the



FIG. 1. A composition space for amphiboles based upon occupancies of the A, B and T sites. Each point of the space corresponding to one or more amphibole end-members is labeled with the name of an end member occurring at that point. Many of these points correspond to more than one end-member. The point labeled "riebeckite", for example, corresponds to the end-members magnesio-riebeckite, glaucophane and ferro-glaucophane, as well as riebeckite. The planes edenite–tremolite–richterite ($Si_T + A + c_B = 12$) and riebeckite–barroisite–tschermakite–taramite–nyboite ($Si_T + A + c_B = 10$) mark the boundaries of amphibole composition space. On the left is shown the composition space for iron–magnesium amphiboles, which lie on the left-hand face of the polyhedron. These end members are obtained by substitution of (Fe,Mg) for Ca on this face.

occupancy of the A site, the average ionic charge on the B site, and the average number of Si atoms in the T site. According to the original IMA classification (Leake 1978), this space is divided into four subspaces by the planes $Si_T = 7$, and charge in the B site = 3. The planes bounding possible amphibole compositions are $A + c_B + Si_T = 10$ and $A + c_B + Si_T = 12$, where A is the occupancy of the A site (0 to 1), c_B, the charge in the B site (6 to 8). (Cases the number of Si atoms in the T site (6 to 8). (Cases in which Si_T is less than 6 are treated in the Discussion section). Intersection of these planes with the bounding faces of Figure 1 defines three subspaces with 7 vertices and one with 4 vertices, giving 13 points representing possible amphibole compositions. Each of these points in Figure 1 is labeled with the name of an IMA end-member lying at that point, although other end members also may lie at the same point, as shown in Table 1. These 13 points include all possible end-members within the space defined by Figure 1; given the occupancies of the A, B and T sites, the numbers of divalent and trivalent atoms in the C sites are fixed by charge-balance considerations. All IMA end members with Si_T greater than or equal to 6 contain two (negative) charges in the OH site, with the exception of kaersutite. Kaersutite contains tetravalent Ti replacing a trivalent ion in the C site, with the charge balanced by replacement of a monovalent anion by a (divalent) oxygen atom in the OH site. In the representation in Figure 1, it therefore plots at "pargasite". The 13 points of Figure 1 may be correlated with IMA end-member nomenclature by considering within-site substitutions. For example, exchange of Fe for Mg in the

C site at the "tremolite" point produces ferro-actinolite, whereas exchange of Mg for Ca in the *B* site produces magnesio-anthophyllite. (In this chemical representation, no notice is taken of structure.) A list of the names so produced is shown in Table 1.

The subspaces of Figure 1 fall into tetrahedral domains whose orientation is uniquely fixed by planes bounding amphibole composition space. Each of the subspaces with 7 vertices may be subdivided into 4 tetrahedra, three with an orthogonal corner, the other bounded by equilateral triangles (Fig. 2). The total composition space of the amphibole group in Figure 1 can therefore be decomposed into 13 tetrahedral domains. Any composition that can be represented within this composition space falls within one of these 13 domains. I assume that a composition contains only end members that are vertices of the domain within which it falls. (A composition could be reduced to mol fractions of more distant end-members.) The net effect of the assumption is to limit the calculated mol fractions to related compositions. Within a given domain, mol fractions can be readily assigned. Suppose a composition lies in the tetrahedron 1-4-5-9 of Figure 2 (at the upper left). Only vertex 5 has a value c_B less than 4. Thus the mol fraction of vertex 5 is $x_5 = 4 - c_B$. Similarly $x_9 = A$, and $x_1 = Si_T - 7$ in order to fulfill constraints on the *A* and *T* sites, and $x_4 = 1 - x_1 - x_5 - x_9$. The condition that a composition lies in this domain at all is easily seen to be $0 < A + (4 - c_B) + (Si_T - 7) < 1$ (*i.e.*, it must lie in the subspace with $Si_T > 7$ and $c_B > 3$, but beneath the plane 1-5-9). Analogous relations hold for the other orthogonal tetrahedra. For equilateral tetrahedra like 1-5-7-9, relations are only slightly more complex: $(x_7 + x_9) = A$, $(x_5 + x_7) = (c_B - 3)$, $(x_1 + x_7) = (Si_T - 7)$ and $(x_1 + x_5) = (1 - A)$, which together give $x_1 = \{(Si_T - 7) + (1 - A) - (c_B - 3)\}/2$, $x_7 = \{(Si_T - 7) - (1 - A) + (c_B - 3)\}/2$, $x_5 = \{(1 - A) + (c_B - 3) - (Si_T - 7)\}/2$ and $x_9 = 1 - \{(Si_T - 7) + (1 - A) + (c_B - 3)\}/2$. Analogous relations hold for the other equilateral tetrahedra.

A hand calculation of a representative composition is shown in detail in Table 2, and a computer program, written in BASIC, which starts from an amphibole composition and produces mol fractions of end members, is listed in the Appendix. This program first makes site assignments, and then uses these assignments to calculate mol fractions.

Site Assignments

The above scheme assumes assignments of analyzed species to crystal chemical sites. As noted by Hawthorne (1983) and many others, this assignment cannot be made in a totally unambiguous fashion. I have followed the recommendation of Leake (1978) and reduced the sum of (O + OH + F + Cl) to 24 where data are available for Fe^{3+}, OH, F and Cl. The problem of reducing results of electron-microprobe analyses lacking H_2O and Fe^{3+} determinations was discussed by Hawthorne (1983) and, more recently, by Richard & Clarke (1990). No procedure is completely satisfactory. A common practice is to calculate the composition on the basis of 46 charges, and then adjust Fe^{2+}/Fe^{3+} to reduce the sum of cations less Ca + Na + K to 13 for cases in which Na + Ca exceeds 1.34, and the sum of cations less Na + K to 15 for cases in which Na + Ca is less than 1.34. In some cases, this procedure produces cation sums less than 15 or greater than 16, as noted by Rock & Leake (1983). It also conflicts with certain end-members as defined by the IMA, since end-member kaersutite has 47 cationic charges rather than 46, as in all other end members. The method here adopted is to normalize electron-microprobe data to 46 + Ti charges, where Ti is the number of Ti atoms in the calculated cell. Fe^{2+}/Fe^{3+} is then adjusted as noted above. Sites are filled in all cases in the order *T, C, B, A* to a maxi-

TABLE 1. AMPHIBOLE END-MEMBERS OBTAINED BY WITHIN-SITE EXCHANGES

(A)(B)(C)(T)	Exchange (site)	End-member @
tremolite	Mg - Fe (C)	ferro-actinolite
()(Ca_2)(Mg_5)(Si_8)	Ca - Fe (B)	ferro-anthophyllite
	Fe - Mg (B)	magnesio-anthophyllite
	Mg - Mn (B)	tirodite
	Mg - Fe (C)	dannemorite
winchite*	Mg - Fe_{3+}(C)	ferro-alumino
()(CaNa)(Mg_4Al)(Si_8)	Al - Fe^{3+} (C)	ferro-ferri
	Fe - Mg (C)	magnesio-ferri
riebeckite	Fe_{3+} - Mg (C)	magnesio-riebeckite
()(Na_2)($Fe^{3+}_2Fe^{2+}_3$)(Si_8)	Fe^{3+} - Al (C)	ferro-glaucophane
	Fe - Mg (C)	glaucophane
	Na - Li (B)	holmquistite
hornblende*	Mg - Fe_{3+}(C)	ferro-alumino
()(Ca_2)(Mg_4Al)(Si_7Al)	Al - Fe^{3+} (C)	ferro-ferri#
	Fe - Mg (C)	magnesio-ferri#
barroisite*	Mg - Fe_{3+}(C)	ferro-alumino
()(CaNa)(Mg_3Al_2)(Si_7Al)	Al - Fe^{3+} (C)	ferro-ferri
	Fe - Mg (C)	magnesio-ferri
tschermakite*	Mg - Fe_{3+}(C)	ferro-alumino
()(Ca_2)(Mg_3Al_2)(Si_6Al_2)	Al - Fe^{3+} (C)	ferro-ferri
	Fe - Mg (C)	magnesio-ferri
richterite	Mg - Fe (C)	ferro-richterite
(Na)(CaNa)(Mg_5)(Si_8)		
arfvedsonite	Fe_{3+} - Mg (C)	magnesio-arfvedsonite
(Na)(Na_2)($Fe^{3+}Fe^{2+}_4$)(Si_8)	Fe^{3+} - Al (C)	eckermannite
	Mg - Fe (C)	ferro-eckermannite
	Fe - Mn (C)	kozulite
edenite	Mg - Fe (C)	ferro-edenite
(Na)(Ca_2)(Mg_5)(Si_8)		
kataphorite*	Mg - Fe_{3+}(C)	ferro-alumino
(Na)(CaNa)(Mg_4Al)(Si_7Al)	Al - Fe^{3+} (C)	ferro-ferri
	Fe - Mg (C)	magnesio-ferri
nyboite*	Mg - Fe_{3+}(C)	ferro-alumino#
(Na)(Na_2)(Mg_3Al_2)(Si_7Al)	Al - Fe^{3+} (C)	ferro-ferri#
	Fe - Mg (C)	magnesio-ferri#
pargasite	Mg - Fe_{3+}(C)	ferro-pargasite
(Na)(Ca_2)(Mg_4Al)(Si_6Al_2)	Al - Fe^{3+} (C)	hastingsite
	Fe - Mg (C)	magnesio-hastingsite
taramite*	Mg - Fe_{3+}(C)	ferro-alumino#
(Na)(Na_2)(Mg_3Al_2)(Si_6Al_2)	Al - Fe^{3+} (C)	ferro-ferri#
	Fe - Mg (C)	magnesio-ferri#

@ where no varietal name given, it is the same as in the left-hand column. * the magnesio-alumino end-member is assumed. # not specified but implied by IMA nomenclature.



FIG. 2. Amphibole subspaces defined by IMA nomenclature. The upper left, upper right and lower left groups of four tetrahedra each represent an "exploded" cube missing one tetrahedral corner (compare Fig. 1). Every composition that can be represented within Figure 1 must fall within one of these 13 tetrahedra. Numbering of the vertices corresponds to vertex names in Figure 1, as follows: 1 tremolite, 2 winchite, 3 riebeckite, 4 hornblende, 5 barroisite, 6 tschermakite, 7 richterite, 8 arfvedsonite, 9 edenite, 10 katophorite, 11 nyboite, 12 pargasite, and 13 taramite.

mum of 8, 5, 2 and 1 atoms, respectively, assuming that (i) Ca cannot be accommodated in the *C* or *A* sites, and (ii) that the *B* site must contain 2 atoms, even at the expense of vacancies in *C*. This procedure is certainly imperfect, but an iron-rich composition (Table 3, column 8) in which the H_2O determination is omitted and total Fe given as FeO gives mol fractions similar to those from a complete analysis, although calculated ferrous and ferric iron differ markedly from the values determined. The calculation is therefore relatively insensitive to lack of data on water and ferric iron. The program has, in any case, an option for inputting site occupancies directly, so that other schemes for data reduction can be used. An example of a printout from the program is shown in Table 3. Some results of the computation applied to amphibole data from the literature are shown in Table 4.

DISCUSSION

For most amphiboles, the most abundant calculated component is the name given to the amphibole by the IMA scheme (Table 4, columns 1–5). In some cases the mol fraction of another end-member is comparable in amount to the named end-member (column 3), and in a few cases three or more end-members have similar mol fractions (column 6). The IMA name does not appear among calculated components in Table 2 or Table 4, column 7. The "contradiction" for hornblende results from features of the IMA scheme. For calcic amphiboles, Na_B may be up to 0.67, and the content of the *A* site up to 0.5 atoms. In Figure 1, an amphibole lying on the join hornblende–barroisite would be called hornblende even though it contains 67% of the barroisite end-member. In composition space (Fig. 1), the

composition used in Table 2 lies closer to edenite and barroisite than to hornblende, and this is reflected in the mol fractions. The composition in column 7 of Table 4 was termed "arfvedsonite" by Hawthorne (1983), although (Na + K) in the A site is less than 0.5. According to IMA rules (Hawthorne 1983, Fig. 4, p. 180), the amphibole should be named "riebeckite", and this is reflected by the computed end-members.

The composition space shown in Figure 1 does not span the possible range of amphibole compositions, since compositions with as few as 4.5 atoms of Si per unit cell are well documented. Shimazaki *et al.* (1984) published results of analyses for which the sum of $A + c_B + \mathrm{Si}_T$ approaches 8. If the plane $A + c_B + \mathrm{Si}_T = 8$ were plotted on Figure 1, it would increase the number of possible amphibole points from 13 to 24, but otherwise introduce no difficulty in the calculation. However, few end-members lying below $A + c_B + \mathrm{Si}_T = 10$ have been defined and named. This region is therefore not included in the calculation. One end-member lying on the plane $A + c_B + \mathrm{Si}_T = 10$ and containing less than 6 atoms of Si per formula unit is $A_1B_2^4C_5^{12}\mathrm{Si}_5\mathrm{Al}_3\mathrm{O}_{22}(\mathrm{OH})_2$ (Bunch & Okrusch 1973). A sample calculation assuming this member is shown in Table 4, column 9. In this formula, the superscripts on B and C indicate the total charge at these positions.

Since the occupancy of the C site is not specifically considered in the calculation, differing occupancies of the C site, including unit formulae with an occupancy of the C site less than 5, are included in the end members as now calculated, from which other end members differing only in occupancy of the C site could be separated if the configuration of the C site is specified. In the proposed calculation, no account is taken of occupancy of the OH site, but end members of chlorine- and fluorine-rich amphiboles are readily calculated if the relevant data are available.

The calculation assumes no vacancies in the B site, and a sum for $A + B + C + T$ (including vacancies in C and T) between 15 and 16. Of 103 compositions calculated from the literature, 18 did not meet these criteria. In many cases, results of the analyses can be forced to meet the above criteria by adjusting the basis to which the analytical data are reduced to less than 24(O + OH + F + Cl). This amounts to an assumption of vacancies in the OH site. The computer program supplies a mechanism for such an adjustment, together with a caution about its use. An example of such a recalculation is shown in Table 4, column 9.

The method used to calculate mol fractions of end-member amphiboles, namely to construct and subdivide a composition space with independent compositional variables, is applicable to other complex solid-solutions, for example chlorite. An alternative

TABLE 2. EXAMPLE OF CALCULATION OF AMPHIBOLE END-MEMBERS

Example: Hawthorne (1983), Appendix C1, anal. 42, p. 365

IMA name: "magnesio-hornblende"

		Sites		Mol fractions in sites (compare Fig. 2)	
SiO_2	48.40	Si_z	6.73		
TiO_2	0.33	Al_z	1.27	$Si_7 = Si_z - 6 =$	0.73
Al_2O_3	11.54			$Si_6 = 7 - Si_z =$	0.27
Fe_2O_3	1.44	Al_c	0.62		
FeO	3.59	Ti_c	0.03	$Ca_2 = c_B - 3 =$	0.59
MnO	0.07	Fe^{3+}_c	0.15	$CaNa = 4 - c_B =$	0.41
MgO	18.03	$*FM_c$	4.16		
CaO	10.70	v_c	0.04	$A = Na_A + K =$	0.37
Na_2O	2.52			$1-A =$	0.63
K_2O	0.60	Ca_B	1.59		
H_2O	2.63	Na_B	0.41	* includes Mg 3.73,	
F	-			Fe^{2+} 0.42, Mn 0.01	
Total	100.23	Na_A	0.27		
		K_A	0.10		

Subspace: edenite-hornblende-barroisite-katophorite-taramite-pargasite-tschermakite (Fig. 1)
Tetrahedron: edenite-barroisite-taramite-tschermakite (5-6-9-13, Fig. 2)

$Mg/(Mg+Mn+Fe) = 3.73/4.16 = 0.897$
$Fe^{3+}/(Fe^{3+}+Al) = 0.15/0.77 = 0.195$
$Fe/(Mg+Mn+Fe) = 0.42/4.16 = 0.101$
$Al/(Fe^{3+}+Al) = 0.62/0.77 = 0.805$

End-members				magnesio	ferro
edenite	$(Si_7-1+A+Ca_2)/2$	0.345		0.309	0.035
tschermakite	$(1-A+Ca_2-Si_7)/2$	0.245	alumino	0.177	0.022
			ferri	0.043	0.002
barroisite	$(Si_7+1-A-Ca_2)/2$	0.385	alumino	0.278	0.031
			ferri	0.067	0.008
taramite	$(1+A-Si_7-Ca_2)/2$	0.025	alumino	0.018	0.003
			ferri	0.003	0.000

Magnesio members found by multiplying end-members by Mg/(Mg+Mn+Fe), magnesio-alumino members by [Mg/(Mg+Mn+Fe)]* [Al/(Fe^{3+}+Al)], with analogous formulae for ferro and ferro-ferri members. Note that hornblende does not appear among the end-members, although it is the name given in the I.M.A. classification. This amphibole's composition lies close to the common face between the tetrahedra hornblende-barroisite-edenite-tschermakite and edenite-tschermakite-barroisite-taramite. In composition space, it lies closer to both barroisite and edenite than to hornblende.

representation is the exchange-vector notation introduced by Thompson (1981), although this method does not give mol fractions of end members, at least not without further assumptions. These mol fractions have thermodynamic value, but caution must be exercised in using them where solution properties are involved. The standard free energy of species i is

$$G_i = G_i^\circ + \mathrm{RT} \ln a_i$$

where G_i° is the standard free energy of pure i at P and T, and a_i is the activity of i. A popular model (Temkin model) for ideal solid-solution assumes a_i to be the product of x_z^n where x is the mol fraction of component of i in z, and n is the total number of atoms in z. Applied to complex solutions like amphiboles, such a model commonly gives too large a correction to the free energy, so that mixing on one or more sites, for example T, may be arbitrarily ignored. The assumption $a_i = x_i$, where x_i is the mol fraction as calculated in this paper, would give

the minimum correction by assuming mixing of units consisting of end-member formulae. The correct value for an ideal solution must lie between these extremes.

TABLE 3. EXAMPLE OF PRINTOUT FROM PROGRAM CLASAMPH

```
hornblende (Hawthorne (1983) Appendix C1 analysis 42)
        Oxide          Cell
SiO2    48.400  Si    6.741
TiO2     0.330  Ti    0.035
Al2O3   11.540  Al    1.894
Fe2O3    1.440  Fe3   0.151
FeO      3.590  Fe2   0.418
MnO      0.070  Mn    0.008
MgO     18.030  Mg    3.743
CaO     10.700  Ca    1.597
Na2O     2.520  Na    0.680
K2O      0.600  K     0.107
H2O      2.630  H     2.443

Total         99.850  Ions         15.374
(O+OH+F+Cl)  24.000  Normalised    0.000

Classification of hornblende (Hawthorne (1983) Appendix C1 analysis 42)
T site: 6.741 Si  1.259 Al
C site: 0.635 Al  0.151 Fe3   0.035 Ti  4.170 FM  0.010 Vacant
B site: 0.000 FM  1.597 Ca    0.403 Na
A site: 0.277 Na  0.107 K     0.000 Li

A site:  Full   0.384  Empty 0.616
B site:  Ca2    0.597  CaNa  0.403
T site:  Si7    0.741  Si6   0.259

0.380 barroisite          0.276 magnesio-alumino  0.031 ferro-alumino
                          0.066 magnesio-ferri    0.007 ferro-ferri
0.236 tschermakite        0.171 magnesio-alumino  0.019 ferro-alumino
                          0.041 magnesio-ferri    0.005 ferro-ferri
0.361 edenite             0.324 magnesio          0.036 ferro
0.023 taramite            0.017 magnesio-alumino  0.002 ferro-alumino
                          0.004 magnesio-ferri    0.000 ferro-ferri

Fe/FM = 0.100  Mg/FM  = 0.898  Mn/FM = 0.000
Al/M3 = 0.808  Fe3/M3 = 0.192  Cr/M3 = 0.000
K/A   = 0.278  Li     = 0.000
```

ACKNOWLEDGEMENTS

Lynton Jaques, Bureau of Mineral Resources, Canberra, Australia, suggested the problem to me. Barrie Clarke and N.M.S. Rock clarified the presentation with constructive reviews. Edgar Froese and Rob Berman discussed mol fractions with me. Deborah Lemkow drafted the figures.

REFERENCES

BUNCH, T.E. & OKRUSCH, M. (1973): Al-rich pargasite. *Am. Mineral.* **58**, 721-726.

DEER, W.A., HOWIE, R.A. & ZUSSMAN, J. (1966): *An Introduction to the Rock-Forming Minerals.* Longmans, Green and Company, London.

HAWTHORNE, F.C. (1983): The crystal chemistry of the amphiboles. *Can. Mineral.* **21**, 173-480.

LEAKE, B.E. (1978): Nomenclature of amphiboles. *Can. Mineral.* **16**, 501-520.

RICHARD, L.R. & CLARKE, D.B. (1990): AMPHIBOL: a program for calculating structural formulae and for classifying and plotting chemical analyses of amphiboles. *Am. Mineral.* **75**, 421-423.

ROCK, N.M.S. & LEAKE, B.W. (1984): The International Mineralogical Association amphibole nomenclature scheme: computerization and its consequences. *Mineral. Mag.* **48**, 211-227.

SHIMAZAKI, H., BANNO, M. & OZAWA, T. (1984): Sadanagaite and magnesio-sadanagaite, new silica-poor members of calcic amphibole from Japan. *Am. Mineral.* **69**, 465-471.

THOMPSON, J.B., JR. (1981): An introduction to the mineralogy and petrology of the biopyriboles. *Rev. Mineral.* **9A**, 141-188.

Received February 2, 1990, revised manuscript accepted November 20, 1990.

TABLE 4. CALCULATION OF REPRESENTATIVE AMPHIBOLES INTO END MEMBERS

	1	2	3	4	5	6	7	8	9
SiO_2 wt.%	58.40	44.89	50.45	37.55	59.06	41.36	47.86	47.86	39.90
TiO_2	0.03	0.67	0.14	0.89	0.20	1.95	0.64	0.64	4.65
Al_2O_3	0.57	17.91	1.96	9.90	12.38	12.49	1.69	1.69	14.35
Fe_2O_3	0.58	0.67	17.52	11.89	2.36	4.25	17.95	-	9.60
FeO	7.85	13.31	17.90	21.40	10.84	12.36	19.91	36.06	0.04
MnO	0.27	0.37	1.40	1.25	0.25	0.11	0.71	0.71	0.08
MgO	29.25	18.09	0.05	1.31	8.82	11.45	0.13	0.13	14.52
CaO	0.14	0.40	0.08	7.28	0.21	10.85	1.64	1.64	12.14
Na_2O	0.08	1.45	6.80	4.09	0.11	1.68	6.25	6.25	1.90
K_2O	0.02	0.05	1.48	2.11	0.05	1.50	0.61	0.61	2.31
Li_2O	-	-	-	-	3.25	-	-	-	-
H_2O	2.80	2.02	0.87	1.84	2.16	1.87	1.67	-	0.50
F	-	-	2.58	0.01	0.18	0.05	1.16	-	0.12
Total	99.99	99.83	101.23	99.92	99.87	99.92	100.22	95.59	100.11
Site occupancies									
Si_T	7.853	6.329	8.000	6.109	7.897	6.186	7.599	7.654	5.851
Al_T	0.090	1.671		1.891	0.103	1.814	0.316	0.319	2.149
Al_C		1.305	0.367	0.007	1.848	0.387			0.331
Ti_C	0.003	0.071	0.017	0.109	0.020	0.219	0.076	0.077	0.513
Fe^{3+}_C	0.002	0.071	2.094	1.456	0.237	0.478	2.060	1.575	1.059
$*FM_C$	4.818	3.553	2.509	3.402	2.805	3.915	2.770	3.340	3.097
$*FM_B$	1.959	1.863	0.068		0.194	0.197			0.093
Ca_B	0.020	0.060	0.014	1.269	0.030	1.739	0.279	0.281	1.907
Na_B	0.021	0.076	1.918	0.731	1.748"	0.064	1.721	1.719	
Na_A		0.320	0.176	0.559		0.423	0.203	0.219	0.540
K_A	0.003	0.009	0.300	0.438	0.009	0.286	0.124	0.128	0.432

- not determined. *FM = Fe^{2+}+Mn+Mg. " = Li

End-members									
tremolite									
winchite					0.015				
riebeckite[1]			0.524				0.497	0.515	
hornblende									
barroisite	0.020	0.030			0.014		0.176	0.142	
tschermakite				0.003		0.260			0.026
richterite			0.082		0.001		0.103	0.139	
arfvedsonite[2]			0.394						
edenite				0.109		0.166			
katophorite									
nyboite		0.008					0.224	0.204	
pargasite[3]				0.048		0.192			0.266
kaersutite				0.109		0.219			0.513
taramite		0.030		0.731		0.064			
anthophyllite	0.914				0.053				
gedrite	0.062	0.641			0.045	0.031			0.002
sodium anth.	0.003	0.291				0.020			
sodium gedrite						0.048			0.045
holmquistite					0.866				0.149*

[1] includes glaucophane; [2] includes eckermannite; [3] includes hastingsite.
* Hypothetical $A_1B_2C_5Si_8Al_3O_{22}(OH)_2$ end-member, see text for discussion.
Column 1: Anthophyllite, Deer et al. (1966, Table 15, anal. 1), reduced to 24.00 (O+OH+F+Cl). C-site occupancy includes 0.177 vacant. 2: Gedrite, Deer et al. (1966, Table 15, anal. 2), reduced to 24.00 (O+OH+F+Cl). 3: Fluor-riebeckite, Hawthorne (1983, Appendix B1, anal. 68), reduced to 24.00 (O+OH+F+Cl). 4: Potassian ferri-taramite, Hawthorne (1983, Appendix B1, anal. 51), reduced to 24.00 (O+OH+F+Cl). C-site occupancy includes 0.026 vacant. 5: Holmquistite, Hawthorne (1983, Appendix C1, anal. 31), reduced to 24.00 (O+OH+F+Cl). C-site occupancy includes 0.090 vacant. 6: Potassian tschermakite, Hawthorne (1983, Appendix B1, anal. 60), reduced to 24.00 (O+OH+F+Cl). 7: Arfvedsonite, Hawthorne (1983, Appendix B1, anal. 64), reduced to 24.00 (O+OH+F+Cl). 8: Arfvedsonite, same in 7, but all iron quoted as FeO, and H_2O and F data removed to simulate the results of an electron-microprobe analysis. " Reduced to 46+Ti cationic charges. 9: Potassian oxy-kaersutite, Hawthorne (1983, Appendix B1, anal. 55), reduced to 23.86 (O+OH+F+Cl) (see text).

APPENDIX: CLASAMPH: a BASIC program to compute mol fractions of amphibole end-members

The following program starts with an amphibole analysis, entered from file, keyboard or a set of site occupancies, and produces an assignment to end-members. A compiled version with source code (in TurboBasic) is available from the authour for $3.00 to cover the cost of computer disks. File analyses for use with CLASAMPH require the structure; analysis title, number of data, name of datum, datum, for example the analysis in Table 3 in a file would have the form:
"hornblende (Hawthorne (1983) Appendix C1 analysis 42)",12
SiO2,48.4,TiO2,0.33,Al2O3,11.54,Fe2O3,1.44,FeO,3.59,MnO,0.07
MgO,18.03,CaO,10.7,Na2O,2.52,K2O,0.6,H2O,2.63,F,0.00
Non-chemical data may be present, but will be ignored.

```
1 dim oxide(18),cell(18),molwt(18),charge(18),namox$(18),ion$(18)
2 dim inox(18),inam$(18),amph$(23),am(20),xm$(20,4):data SiO2,Si,60.084,4
3 data TiO2,Ti,79.899,4,ZrO2,Zr,123.219,4,Al2O3,Al,50.981,3,Fe2O3,Fe3,79.846,3
4 data Cr2O3,Cr,75.995,3,FeO,Fe2,71.846,2,MnO,Mn,70.937,2,NiO,Ni,74.71,2
5 data MgO,Mg,40.304,2,ZnO,Zn,71.001,2,Li2O,Li,14.941,1,CaO,Ca,56.079,2
6 data Na2O,Na,30.989,1,K2O,K,47.098,1,H2O,H,9.01,1,Cl,Cl,35.45,0,F,F,19.,0
7 cls:for i=1 to 18:read namox$(i),ion$(i),molwt(i),charge(i):next i
8 data A0B4Si8,winchite,A0B2Si8,hornblende,barroisite,tschermakite,richterite
9 data A1B2Si8,edenite,kataphorite,nyboite,A1B4Si6,taramite,anthophyllite
10 data gedrite,Na-anthophyllite,Na-gedrite,holmquistite,kaersutite,A1B4Si5
11 for i=1 to 20:read amph$(i):xm$(i,4)="ferro-ferri":xm$(i,2)="ferro-alumino"
12 xm$(i,1)="magnesio-alumino":xm$(i,3)="magnesio-ferri":next i
13 xm$(1,1)="tremolite":xm$(8,3)="arfvedsonite":xm$(7,2)="ferro"
14 xm$(9,1)="magnesio":xm$(9,2)="ferro":xm$(8,4)="Mg arfvedsonite"
15 for i=14 to 18:xm$(i,1)="magnesio":xm$(i,2)="ferro":next i
16 xm$(3,2)="Fe glaucophane":xm$(3,3)="Mg riebeckite":xm$(3,4)="riebeckite"
17 xm$(12,1)="pargasite":xm$(12,2)="Fe pargasite":xm$(12,3)="Mg hastingsite"
18 xm$(12,4)="hastingsite":xm$(1,2)="ferro-actinolite":xm$(7,1)="magnesio"
19 xm$(8,1)="eckermannite":xm$(8,2)="Fe eckermannite":xm$(3,1)="glaucophane"
20 print"CLASAMPH computes mol fractions of amphibole end members from one of"
21 print"complete analysis, electron probe data, or site occupancies.":print
22 input"Return to DOS (q) or enter site occupancies (s) or continue (c)? ",a$
23 if a$="q" or a$="Q" then goto 264
24 st=0:if a$="s" or a$="S" then st=1:goto 293
' Data input and validation
25 print:input"Data from file (f) or keyboard (k) ",a$
26 f$="":if a$<>"f" and a$<>"F" and a$<>"k" and a$<>"K" then goto 25
27 if a$="F" or a$="f" then goto 265
28 cls:input "Enter mineral identifier ",minername$
29 print:print "Input analysis":atot=0
30 for i=1 to 18:print using "\    \";namox$(i),:input,oxide(i)
31 atot=atot+oxide(i):next i
32 if atot<90 then print"Data defective or incomplete":goto 22
33 cls:print minername$:print:for i=1 to 18
34 if oxide(i)>.001 then print using" \     \  ##.###";namox$(i),oxide(i)
35 next i:print:print using" Total   ###.##";atot:print
36 input"OK to process? (y/n) ",a$:if a$="n" or a$="N" and f$<>"" then goto 260
37 if a$="N" or a$="n" then cls:goto 22
38 char=0:btot=0:input"Normalise total to fixed value? (y/n) ",a$
39 q=1.e-6:m=18:if a$<>"y" then goto 42
40 input"Enter value for normalisation ",btot:rat=btot/atot
41 for i=1 to m:oxide(i)=oxide(i)*rat:next i
42 for i=1 to m:cell(i)=0:cell(i)=oxide(i)/molwt(i):char=char+cell(i)*charge(i)
43 next i:zb=0:if oxide(16)>.1 then zb=1
' Reduction to standard cell
44 basis=46+(1-zb)*cell(2)+2*zb
45 for i=1 to m:cell(i)=cell(i)*basis/char:next i
' Calculation of Fe3/Fe2 if required
46 number2=0:fe2=0:fe3=0:revtot=0:pm=m:vf=0:if cell(5)>1.e-6 then goto 58
47 num=cell(13)+cell(14)+cell(15):number2=13:if num<1.34 then number2=15
48 for i=1 to 12:vf=vf+cell(i):next i:if num<1.34 then vf=vf+cell(13)
49 if number2/vf>1 then goto 56
50 if number2/vf>1/(1+cell(7)/basis) then goto 53
```

```
51 for i=1 to pm:cell(i)=cell(i)/(1+cell(7)/basis):next i
52 cell(5)=cell(7):cell(7)=0:goto 56
53 for i=1 to pm:cell(i)=cell(i)*number2/vf:next i
54 cell(5)=cell(5)+(1-number2/vf)*basis
55 cell(7)=cell(7)-(1-number2/vf)*basis
56 fe2=oxide(7)*cell(7)/(cell(7)+cell(5)):fe3=(oxide(7)-fe2)*79.846/71.846
57 revtot=atot+fe2+fe3-oxide(7):if btot>0 then revtot=btot+fe2+fe3-oxide(7)
58 n3=0:for i=1 to 15:n3=n3+cell(i):next i
' Assignment of standard cell to sites
59 anerr=0:siz=cell(1):if siz>8 then siz=8
60 alz=8-siz:if alz>cell(4) then alz=cell(4)
61 fez=8-siz-alz:if fez>cell(5) then fez=cell(5)
62 tiz=8-siz-alz-fez:if tiz>cell(2)+cell(3) then tiz=cell(2)+cell(3)
63 defz=8-siz-alz-fez-tiz:if siz<5.50 then anerr=-1
64 ca=cell(13):na=cell(14):k=cell(15)
65 siy=cell(1)-siz:aly=cell(4)-alz:fey=cell(5)+cell(6)-fez
66 tiy=cell(2)+cell(3)-tiz:fm=0:for i=7 to 12:fm=fm+cell(i):next i
67 fmy=5-siy-aly-fey-tiy:if fmy>fm then fmy=fm
68 defy=5-siy-aly-fey-tiy-fmy:fmx=fm-fmy:cax=ca
69 nax=2-fmx-cax:if nax<0 then nax=0
70 if nax>na then nax=na
71 naa=na-nax:a=naa+k:m3=fey+aly+q
72 cs=2-fmx-cax-nax:if cs>0 then fmx=fmx+cs:fmy=fmy-cs:defy=defy+cs
73 xb=0:if fmx+cax-2>0 then xb=fmx+cax-2:anerr=1
74 xa=0:if naa+k-1>0 then xa=naa+k-1:anerr=2
75 lix=cell(12):if lix>fmx then lix=fmx
76 fmx=fmx-lix:fm2=fmx/2:cana=fmx+cax:if nax+lix<cana then cana=nax+lix
77 ca2=(fmx+cax-cana)/2:na2=(nax+lix-cana)/2:cb=2*na2+3*cana+4*ca2:fm=fm-lix
78 q=1.e-6:si8=0:si7=0:si6=0:si5=0:if siz>7 then si8=siz-7:si7=1-si8
79 if siz<7 and siz>6 then si7=siz-6:si6=1-si7
80 if siz<6 then si6=siz-5:si5=1-si6
81 stot=a+siz+cb-12:if stot>q then anerr=3
82 if stot<-2 then anerr=-2
83 if anerr<>0 then goto 134
' Classification of amphibole
84 for i=1 to 20:am(i)=0:next i:if si8=0 then goto 102
85 if na2>0 then goto 94
' tremolite-winchite-richterite-edenite-hornblende-barroisite-katophorite
86 if ca2-a+si8>1 then goto 88
' edenite-richterite-barroisite-katophorite
87 am(9)=ca2:am(5)=1-a:am(7)=si8:am(10)=1-am(9)-am(5)-am(7):goto 114
88 if cana+a+si8>1 then goto 90
' barroisite-edenite-hornblende-tremolite
89 am(5)=cana:am(9)=a:am(1)=si8:am(4)=1-am(5)-am(9)-am(1):goto 114
90 if ca2+a+si7>1 then goto 92
' tremolite-barroisite-richterite-winchite
91 am(1)=ca2:am(5)=si7:am(7)=a:am(2)=1-am(1)-am(5)-am(7):goto 114
' tremolite-richterite-barroisite-edenite
92 am(1)=(ca2-a+si8)/2:am(7)=(a+si8-ca2)/2:am(5)=(2-a-ca2-si8)/2
93 am(9)=(a+ca2-si8)/2:goto 114
' winchite-riebeckite-arfvedsonite-richterite-katophorite-barroisite-nyboite
94 if ca2+1-a+si7>1 then goto 96
' richterite-riebeckite-nyboite-arfvedsonite
95 am(7)=cana:am(3)=1-a:am(11)=si7:am(8)=1-am(7)-am(3)-am(11):goto 114
96 if na2+1-a+si8>1 then goto 98
' nyboite-barroisite-richterite-katophorite
97 am(11)=na2:am(5)=1-a:am(7)=si8:am(10)=1-am(11)-am(5)-am(7):goto 114
98 if na2+a+si7>1 then goto 100
' riebeckite-richterite-barroisite-winchite
99 am(3)=na2:am(7)=a:am(5)=si7:am(2)=1-am(7)-am(3)-am(5):goto 114
' richterite-riebeckite-barroisite-nyboite
100 am(7)=(si8-na2+a)/2:am(3)=(na2-a+si8)/2:am(5)=(2-a-na2-si8)/2
101 am(11)=(a+na2-si8)/2:goto 114
```

```
102 if si5>0 then goto 113
103 if na2>0 then goto 112
' hornblende-barroisite-katophorite-edenite-pargasite-taramite-tschermakite
104 if cana+a+si6>1 then goto 106
' barroisite-tschermakite-edenite-hornblende
105 am(5)=cana:am(6)=si6:am(9)=a:am(4)=1-am(5)-am(6)-am(9):goto 114
106 if cana+1-a+si7>1 then goto 108
' taramite-tschermakite-edenite-pargasite
107 am(13)=cana:am(6)=1-a:am(9)=si7:am(12)=1-am(13)-am(6)-am(9):goto 114
108 if ca2+1-a+si6>1 then goto 110
' edenite-barroisite-taramite-katophorite
109 am(9)=ca2:am(5)=1-a:am(13)=si6:am(10)=1-am(9)-am(5)-am(13):goto 114
' edenite-barroisite-tschermakite-taramite
110 am(9)=(ca2+a-si6)/2:am(5)=(2-a-si6-ca2)/2:am(6)=(ca2+si6-a)/2
111 am(13)=(a+si6-ca2)/2:goto 114
' nyboite-barroisite-taramite-katophorite
112 am(11)=na2:am(5)=1-a:am(13)=si6:am(10)=1-am(11)-am(5)-am(13):goto 114
' taramite-tschermakite-pargasite-A1B4Si5
113 am(13)=cana:am(6)=1-a:am(20)=si5:am(12)=1-am(13)-am(6)-am(20)
114 a1=am(1)+am(4)+am(6)+am(9)+am(12):rat=fm2/(a1+q):if rat>1 then rat=1
' Conversion of Ca2 amphibole to Fe Mg amphibole
115 am(14)=rat*(am(1)+am(4)/2):am(15)=rat*(am(6)+am(4)/2):am(16)=rat*am(9)
116 am(17)=rat*am(12):rt=1-rat:am(1)=am(1)*rt:am(4)=am(4)*rt:am(6)=am(6)*rt
117 am(9)=am(9)*rt:am(12)=am(12)*rt:ca2=ca2*rt:rs=fm2-rat*a1
118 a1=am(13):if am(10)<a1 then a1=am(10)
119 a2=am(13)-a1:if am(5)<a2 then a2=am(5)
120 a3=am(7):if am(10)-a1<a3 then a3=am(10)-a1
121 a4=am(7)-a3:if am(5)-a2<a4 then a4=am(5)-a2
122 a5=am(7)-a3-a4:if am(2)<a5 then a5=am(2)
123 a6=am(10)-a1-a3:a7=am(5)-a2-a4:a8=am(2)-a5
124 tf=a1+a2+a3+a4+a5+a6/2+a7/2+a8/2:z=rs/(tf+q):if z>1 then z=1
' Conversion of CaNa amphibole to Fe Mg amphibole
125 am(15)=am(15)+z*(a2+a4/2+a7/2):am(16)=am(16)+z*(a3+a6/2)
126 am(17)=am(17)+z*a1:am(14)=am(14)+z*(a5+a8/2+a4/2):am(2)=am(2)-z*(a5+a8)
127 am(3)=am(3)+z*(a7/2+a8/2):am(8)=am(8)+z*(a3+a5):am(5)=am(5)-z*(a2+a4+a7)
128 am(11)=am(11)+z*(a1+a2+a6/2):cana=cana-z*2*tf:na2=na2+z*tf
129 am(10)=am(10)-z*(a1+a3+a6):am(13)=am(13)-z*(a1+a2):am(7)=am(7)-z*(a3+a4+a5)
130 am(18)=lix/2:if lix/2>am(3) then am(18)=am(3) ' holmquistite
131 am(3)=am(3)-am(18)
132 am(19)=tiy:if tiy>am(12) then am(19)=am(12) ' kaersutite
133 am(12)=am(12)-am(19)
' Print results to screen
134 cls:print minername$:print:f=1:if st=1 then goto 148
135 if num<1.8 and num>1.1 Then print"WARNING:1.1<Na+Ca+K<1.8.Analysis dubious"
136 print "        Oxide        Cell"
137 for j=1 to 18:if oxide(j)+cell(j)<.001 then goto 139
138 print using"\     \ ##.###  \  \ ##.###";namox$(j),oxide(j),ion$(j),cell(j)
139 next j:ztot=0:for i=1 to pm:ztot=ztot+cell(i)*charge(i)/2:next i
140 print "":if btot=0 then print using "Total  ###.### Ions ##.### ";atot,n3
141 if btot>0 then print using "Total ###.### Ions ##.### ";btot,n3
142 print"(O+OH+F+Cl) ";:print using"##.###  Normalised ##.###";ztot,number2
143 if basis<>48 and zb=1 then print "CAUTION: (O+OH+F+Cl) adjusted!"
144 if btot=0 then goto 146
145 print using"Analysis normalised to ###.## using factor #.###";btot,rat
146 if fe3=0 then goto 148
147 print using"Normalised FeO=##.## Fe2O3=##.## Total=###.##";fe2,fe3,revtot
148 print:print"Site occupancies"
149 q=.0001:print using"T site: #.### Si  #.### Al";siz,alz,
150 if fez>q then print using"  #.### Fe3";fez,
151 if tiz>q then print using"  #.### Ti  #.### Vacant";tiz,
152 if defz>q then print using"  #.### Vacant";defz,
153 print:print using"C site: #.### Al  #.### Fe3  #.### Ti";aly,fey,tiy,
154 print using"  #.### FM";fmy,
```

```
155 if siy>=q then print using" #.### Si ";siy,
156 if defy>=q then print using" #.### Vacant";defy,
157 print:print using"B site: #.### FM  #.### Ca  #.### Na";fmx,cax,nax,
158 if cell(12)>q then print using" #.### Li";cell(12),
159 print:print using"A site: #.### Na  #.### K";naa,k:print
160 if anerr<>0 then goto 281
161 input"To print amphibole classification strike any key.",A$
162 cls:print"Classification of ";minername$:print:b$=""
163 print using"A site:  Full  #.###  Empty #.###";a,1-a
164 print using"B site:";b$,:if fm2>q then print using"  FeMg  #.###";fm2,
165 if ca2>q then print using"  Ca2   #.###";ca2,
166 if cana>q then print using"  CaNa  #.###";cana,
167 if na2>q then print using"  Na2   #.###";na2,
168 print:print using"T site:";b$,:if si8>q then print using"  Si8   #.###";si8,
169 if si7>q then print using"  Si7   #.###";si7,
170 if si6>q then print using"  Si6   #.###";si6,
171 if si5>q then print using"  Si5   #.###";si5,
172 print:print:z=0.001:for i=1 to 23:if abs(am(i))<z then goto 184
173 print using"#.### \              \   ";am(i),amph$(i),
174 if i>13 then goto 182
175 if i=1 or i=7 or i=9 then goto 182
176 print using"#.### \             \  ";(cell(10)/fm)*(aly/m3)*am(i),xm$(i,1),
177 print using" #.### \              \  ";(cell(7)/fm)*(aly/m3)*am(i),xm$(i,2)
178 ct=(cell(10)/fm)*(fey/m3)*am(i)
179 print using"                          #.### \               \  ";ct,xm$(i,3),
180 print using" #.### \               \ ";(cell(7)/fm)*(fey/m3)*am(i),xm$(i,4)
181 goto 184
182 print using"#.### \              \   ";(cell(10)/fm)*am(i),xm$(i,1),
183 print using"#.### \              \   ";(cell(7)/fm)*am(i),xm$(i,2)
184 next i:print:mg=cell(10):mn=cell(8)/fm
185 print using"Fe/FM=#.### Mg/FM =#.### Mn/FM=#.###";cell(7)/fm,mg/fm,mn
186 print using"Al/M3=#.### Fe3/M3=#.### Cr/M3=#.###";aly/m3,fey/m3,cell(6)/m3
187 print using"K/A  =#.### Li    =#.###";k/(k+na-nax+.000001),cell(12)
' Print results to file
188 print:input"Save this result to file? (y/n) ",a$
189 if a$<>"y" then goto 255
190 if f2$<>"" then goto 192
191 input"Specify path and file name for result file ",f2$
192 open f2$ for append as #2:
193 print #2,:print #2,:print #2, minername$:print:if st=1 then goto 210
194 if num<1.8 and num>1.1 then print #2,"1.1<Na+Ca+K<1.8. Analysis dubious"
195 print #2,"        Oxide        Cell"
196 for j=1 to 18:if oxide(j)+cell(j)<.001 then goto 199
197 print #2, using"\    \ ##.###  ";namox$(j),oxide(j),
198 print #2, using"\  \ #.###";ion$(j),cell(j)
199 next j:ytot=0:ztot=0:for i=1 to pm:ytot=ytot+cell(i)
200 ztot=ztot+cell(i)*charge(i)/2:next i:print #2,
201 if ytot=0 then ytot=n3
202 if btot=0 then print #2, using"Total ###.### Ions ##.### ";atot,n3
203 if btot>0 then print #2, using "Total ###.### Ions ##.### ";btot,n3
204 print #2"(O+OH+F+Cl):print #2, using" ##.### Normalised ##.###";ztot,number2
205 if btot=0 then goto 207
206 print #2, using"Analysis normalised to ###.## using factor #.###";btot,rat
207 if fe3=0 then goto 210
208 print #2, using"Normalised FeO=##.## Fe2O3=##.##";fe2,fe3,
209 print #2, using" total=###.##";revtot
210 print #2,:print #2,"Classification of ";minername$:
211 q=.0001:Print #2, using"T site: #.### Si  #.### Al";siz,alz,
212 if fez>q then print #2, using"   #.### Fe3";fez,
213 if tiz>q then print #2, using"   #.### Ti";tiz,
214 if defz>q then print #2, using"  #.### Vacant";defz,
215 print #2,:print #2, using"C site: #.### Al  #.### Fe3  ";aly,fey,
216 print #2, using"#.### Ti  #.### FM";tiy,fmy,
```

```
217 if siy>=q then print #2, using" #.### Si ";siy,
218 if defy>=q then print #2, using"  #.### Vacant";defy,
219 print #2,:print #2, using"B site: #.### FM  #.### Ca   ";fmx,cax,
220 print #2, using"#.### Na";nax
221 print #2, using"A site: #.### Na  #.### K    #.### Li";naa,k,lix:print #2,
222 print #2, using"A site:  Full  #.###  Empty #.###";a,1-a
223 print #2, using"B site:";b$,
224 if fm2>q then print #2, using"  FeMg  #.###";fm2,
225 if ca2>q then print #2, using"  Ca2   #.###";ca2,
226 if cana>q then print #2, using"  CaNa  #.###";cana,
227 if na2>q then print #2, using"  Na2   #.###";na2,
228 print #2,:print #2, using"T site:";b$,
229 if si8>q then print #2, using"  Si8   #.###";si8,
230 if si7>q then print #2, using"  Si7   #.###";si7,
231 if si6>q then print #2, using"  Si6   #.###";si6,
232 if si5>q then print #2, using"  Si5   #.###";si5,
233 print #2,:print #2,:z=0.001:For i=1 to 23:if abs(am(i))<z then goto 248
234 print #2, using"#.### \              \   ";am(i),amph$(i),
235 if i>13 then goto 246
236 if i=1 or i=7 or i=9 then goto 246
237 az=(cell(10)/fm)*(aly/m3)*am(i)
238 print #2, using"#.### \              \   ";az,xm$(i,1),
239 az=(cell(7)/fm)*(aly/m3)*am(i)
240 print #2, using"#.### \              \   ";az,xm$(i,2)
241 q=(cell(10)/fm)*(fey/m3)*am(i):q$=xm$(i,3)
242 print #2, using"                          #.### \              \   ";q,q$,
243 az=(cell(7)/fm)*(fey/m3)*am(i)
244 print #2, using"#.### \              \   ";Az,xm$(i,4)
245 goto 248
246 print #2, using"#.### \              \   ";(Cell(10)/fm)*am(i),xm$(i,1),
247 print #2, using"#.### \              \   ";(Cell(7)/fm)*am(i),xm$(i,2)
248 next i:print #2,:mg=cell(10):mn=cell(8)/fm
249 print #2, using"Fe/FM = #.###  Mg/FM  = #.###  ";Cell(7)/fm,mg/fm,
250 print #2, using"Mn/FM = #.###";Mn/fm
251 print #2, using"Al/M3 = #.###  Fe3/M3 = #.###  ";Aly/m3,fey/m3,
252 print #2, using"Cr/M3 = #.###";Cell(6)/m3
253 print #2, using"K/A   = #.###  Li    = #.###";K/(k+na-nax+.000001),Cell(12)
254 print #2,:print #2,:print #2,:close #2
' Further calculation options
255 print:input"Process another analysis? (y/n) ",a$
256 if a$="N" or a$="n" then goto 264
257 if st=1 then goto 22
258 if f$="" then goto 25
259 print"Process next analysis from ";f$;"? (y/n)",:input" ",a$
260 if a$="y" or a$="Y" then goto 265
261 print"Process another analysis from ";f$;"? (y/n)",:input" ",a$
262 if a$="n" or a$="N" then goto 25
263 print"Now at analysis ";als%-1,:input"process analysis ";als%:goto 265
264 end
' Read analysis from file
265 for i=1 to 18:oxide(i)=0:next i:on error goto 277
266 if f$<>"" then goto 268
267 print:als%=1:input"Specify drive, path and file name for data ";f$
268 open f$ for input as #1:atot=0:for k=1 to als%:if eof(1)=-1 then goto 279
269 input #1,minername$,m:for i=1 to m:input #1,inam$(i),inox(i):next i
270 next k:for i=1 to m:for j=1 to 18
271 if namox$(j)=inam$(i) then oxide(j)=inox(i):goto 274
272 next j:print inam$(i);" not in data bank. It will be ignored.",
273 input" press any key",a$
274 next i:for i=1 to 18:if oxide(i)<0 then oxide(i)=0
275 atot=atot+oxide(i):next i:if atot<50 then print"No data":goto 22
276 als%=als%+1:close #1:goto 33
277 input"File error; bad name or data. Press any key. ",a$
```

```
278 close #1:f$="":f2$="":resume 22
279 input"End of file encountered. Press any key to continue.",a$
280 close #1:f$="":goto 22
' Options for changing basis to permit classification
281 if anerr=-1 then print"Amphibole must have Si in cell>5.5":goto 255
282 if anerr=-2 then print"Siz+A+Cb<10. Not in amphibole space":goto 255
283 q=1.e-6:if anerr=1 then print "B site overfilled"
284 if anerr=2 then print "A site overfilled"
285 if anerr=3 then print "Siz+A+Cb>12. Not in amphibole space"
286 input"Adjust the basis in attempt to classify?  (y/n) ",a$
287 if a$="n" or a$="N" then goto 255
288 fac=1:if a$="y" and anerr=1 then fac=15/(15+xb+q)
289 if a$="y" and anerr=2 then fac=3/(3+xa+q)
290 if a$="y" and anerr=2 and 1/(k+q)<fac then fac=1/(k+q)
291 if a$="y" and anerr=3 then fac=12/(12+stot)
292 basis=basis*fac:for i=1 to m:cell(i)=cell(i)*fac:next i:goto 58
' Entry of site occupancies
293 cls:print"Site occupancy input":input"Enter analysis name ",minername$
294 input"Enter Si T  ",siz:if siz>5.5 and siz<8.05 then goto 296
295 input"Value must be 8.05<SiT<5.50. Press any key. ",a$:goto 22
296 input"Enter Al T  ",alz:if siz+alz<8.05 then goto 298
297 input"SiT+AlT must be <8.05. Press any key. ",a$:goto 22
298 fez=0:tiz=0:if siz+alz>7.99 then goto 302
299 input"Enter Fe3 T ",fez:input"Enter Ti T  ",tiz
300 if siz+alz+fez+tiz>8.05 then input"T site>8.05. Invalid. ",a$:goto 22
301 defz=8-siz-alz-fez-tiz
302 siy=siz-8:if siy<0 then siy=0
303 siz=siz-siy:input"Enter Ti C  ",tiy:input"Enter Fe3 C ",fey
304 input"Enter Al C  ",aly:input"Enter Mg C  ",mgy:input"Enter Fe2 C ",f2y
305 input"Enter Mn C  ",mny:input"Rest C ions ",oy
306 fmy=mgy+f2y+mny+oy:defy=5-siy-tiy-fey-aly-fmy-oy:m3=fey+aly
307 if defy<-0.001 then input"C site overfilled. Press a key",a$:goto 22
308 input"Enter Mg B  ",mgb:input"Enter Fe B  ",feb
309 input"Enter Mn B  ",mnb:input"Enter Ca B  ",cax:fmx=mgb+feb+mnb+ob
310 input"Enter Na B  ",nax:input"Enter Li B  ",lix:fm=fmx+fmy
311 input"Rest B ions ",ob:totb=mgb+feb+mnb+cax+nax+lix+ob
312 if totb<2.01 and totb>1.99 then goto 314
313 input"Invalid analysis. B>2.01 or <1.99. Press a key",a$:goto 22
314 input"Enter Na A  ",naa:input"Enter K A   ",k:a=naa+k
315 if a>1 then input"A site overfilled. Invalid. Press a key",a$:goto 22
316 itot=4*(siz+tiz+tiy)+3*(alz+aly+fez+fey)+2*(fmx+fmy+cax+oy)+nax+lix+k
317 if itot<47 and itot>45.5 then goto 319
318 input"Invalid sum of cation charges <45.5 or >48. Press a key",a$:goto 22
319 cb=2*(cax+fmx)+lix+nax:fmx=fmx+lix:na=nax+naa:cell(12)=lix
320 cell(10)=mgy+mgb:cell(7)=f2y+feb:cell(8)=mny+mnb:goto 76
```